We consider the electrical circuit containing an LC transmission line described by (5) in page 161 of V. Rasvan, S.-I. Niculescu, *Oscillations in lossless propagation models: a Liapunov-Krasovskii approach*, IMA J. Mathematical Control and Information, 19 (2002), pp. 151-172.

Let us point out that system (5) is a non-linear differential time-delay system.

We are going to show how to obtain a π-flat output for the non-linear system by studying the linearized system by means of *OreModules*.

First, we define the Ore algebra of differential time-delay operators. In particular, we add all the constants of the system in "*comm*". Let us point out that we use the notation $\alpha = \sqrt{\frac{C}{L}}$.

```
> Alg := DefineOreAlgebra(diff=[D,t], dual_shift=[delta,s], polynom=[t,s],
> comm=[R1,R2,C1,C2,alpha], shift_action=[delta,t,sqrt(L*C)]):
```

The non-linear system is defined by:

```
> evalm([[R1*C1*D(v1)(t)+v1(t)+R1*alpha*v1(t)+R1*f1(v1(t))-2*R1*alpha
> *v2(t-(L*C)^(1/2))+E(t)],
> [(1+R2*alpha)*C2*D(v2)(t)+alpha*v2(t)-2*alpha*eta1(t-(L*C)^(1/2))],
> [-v1(t)+eta1(t)+eta2(t-(L*C)^(1/2))],
> [-1/(1+R2*alpha)*v2(t)+(1-R2*alpha)/(1+R2*alpha)*eta1(t-(L*C)^(1/2))+eta2(t)]])
> = matrix([[0], [0], [0], [0]]);
```

$$\begin{bmatrix} R1\,C1\,\mathrm{D}(v1)(t) + \mathrm{v1}(t) + R1\,\alpha\,\mathrm{v1}(t) + R1\,\mathrm{f1}(\mathrm{v1}(t)) - 2\,R1\,\alpha\,\mathrm{v2}(t-\sqrt{L\,C}) + \mathrm{E}(t) \\ (1+R2\,\alpha)\,C2\,\mathrm{D}(v2)(t) + \alpha\,\mathrm{v2}(t) - 2\,\alpha\,\eta1(t-\sqrt{L\,C}) \\ -\mathrm{v1}(t) + \eta1(t) + \eta2(t-\sqrt{L\,C}) \\ -\dfrac{\mathrm{v2}(t)}{1+R2\,\alpha} + \dfrac{(1-R2\,\alpha)\,\eta1(t-\sqrt{L\,C})}{1+R2\,\alpha} + \eta2(t) \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}$$

Let us point out that the non-linearity of the system appears only in the term $R1\ \mathrm{f1}(\mathrm{v1}(t))$ of the first equation. We first consider the system without the non-linear term in the first equation. Therefore, the linear differential time-delay system is defined by:

```
> R :=evalm([[R1*C1*D+1+R1*alpha,-2*R1*alpha*delta,0,0,1],
> [0,(1+R2*alpha)*C2*D+alpha,-2*alpha*delta,0,0],[-1,0,1,delta,0],
> [0,-1/(1+R2*alpha),(1-R2*alpha)/(1+R2*alpha)*delta,1,0]]);
```

$$R := \begin{bmatrix} R1\,C1\,\mathrm{D}+1+R1\,\alpha & -2\,R1\,\alpha\,\delta & 0 & 0 & 1 \\ 0 & (1+R2\,\alpha)\,C2\,\mathrm{D}+\alpha & -2\,\alpha\,\delta & 0 & 0 \\ -1 & 0 & 1 & \delta & 0 \\ 0 & -\dfrac{1}{1+R2\,\alpha} & \dfrac{(1-R2\,\alpha)\,\delta}{1+R2\,\alpha} & 1 & 0 \end{bmatrix}$$

Then, we define *R_ adj* by using an involution of the Ore algebra *Alg*.

```
> R_adj := Involution(R, Alg);
```

$$R\_adj := \begin{bmatrix} -R1\,C1\,\mathrm{D}+1+R1\,\alpha & 0 & -1 & 0 \\ 2\,R1\,\alpha\,\delta & -C2\,\mathrm{D}-C2\,\mathrm{D}\,R2\,\alpha+\alpha & 0 & -\dfrac{1}{1+R2\,\alpha} \\ 0 & 2\,\alpha\,\delta & 1 & -\dfrac{\delta}{1+R2\,\alpha}+\dfrac{\delta\,R2\,\alpha}{1+R2\,\alpha} \\ 0 & 0 & -\delta & 1 \\ 1 & 0 & 0 & 0 \end{bmatrix}$$

Let us check whether or not the linear system is controllable, parametrizable and flat.

```
> st := time(): Ext1 := Exti(R_adj, Alg, 1); time()-st;
```

$$Ext1 := \left[\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix},\right.$$

$$\begin{bmatrix} -1 & 0 & 1 & \delta & 0 \\ R1\,C1\,\mathrm{D} + 1 + R1\,\alpha & -2\,R1\,\alpha\,\delta & 0 & 0 & 1 \\ 0 & 1 & -\delta + \delta\,R2\,\alpha & -1 - R2\,\alpha & 0 \\ 0 & -C2\,\mathrm{D} + C2\,\mathrm{D}\,R2\,\alpha + \alpha & 0 & -2\,\alpha & 0 \end{bmatrix},$$

$$\left[-\delta^2\,\alpha - C2\,\mathrm{D} - C2\,\mathrm{D}\,R2\,\alpha + \mathrm{D}\,C2\,\delta^2 - R2\,\alpha\,\mathrm{D}\,C2\,\delta^2 - \alpha\right]$$
$$[-2\,\alpha\,\delta]$$
$$[-C2\,\mathrm{D} - C2\,\mathrm{D}\,R2\,\alpha - \alpha]$$
$$[-\alpha\,\delta + \mathrm{D}\,C2\,\delta - R2\,\alpha\,\mathrm{D}\,C2\,\delta]$$
$$\Big[\mathrm{D}^2\,C2\,R2\,\alpha\,R1\,C1 - C2\,\mathrm{D}\,\delta^2\,R1\,\alpha + C2\,\mathrm{D}\,R2\,\alpha + \mathrm{D}^2\,C2\,R1\,C1 + \alpha - 3\,\alpha^2\,\delta^2\,R1$$
$$+ \alpha^2\,R1 + C2\,\mathrm{D} + \alpha\,R1\,C1\,\mathrm{D} + \mathrm{D}\,C2\,R2\,\alpha^2\,R1 + \mathrm{D}\,C2\,R1\,\alpha - \mathrm{D}\,C2\,\delta^2 + \delta^2\,\alpha$$
$$- C2\,\mathrm{D}^2\,\delta^2\,R1\,C1 + R2\,\alpha\,\mathrm{D}\,C2\,\delta^2 + C2\,\mathrm{D}\,R2\,\alpha^2\,\delta^2\,R1 + C2\,\mathrm{D}^2\,R2\,\alpha\,\delta^2\,R1\,C1$$
$$\left.+ \alpha\,\delta^2\,R1\,C1\,\mathrm{D}\Big]\right]$$

1.041

As the first matrix *Ext1* [1] of *Ext1* is the identity matrix, we obtain that the *Alg*-module associated with R is torsion-free. Therefore, the linear system is controllable, and thus, parametrizable. A parametrization of the system is then given by the matrix *Ext1* [3] or, equivalently, by:

```
> P := Parametrization(R, Alg):
> v1(t)=P[1,1];
```

$$\mathrm{v1}(t) = -\alpha\,\xi_1(t - 2\,\sqrt{L\,C}) - C2\,\mathrm{D}(\xi_1)(t) - C2\,R2\,\alpha\,\mathrm{D}(\xi_1)(t) + C2\,\mathrm{D}(\xi_1)(t - 2\,\sqrt{L\,C})$$
$$- C2\,R2\,\alpha\,\mathrm{D}(\xi_1)(t - 2\,\sqrt{L\,C}) - \alpha\,\xi_1(t)$$

```
> v2(t)=P[2,1];
```

$$\mathrm{v2}(t) = -2\,\alpha\,\xi_1(t - \sqrt{L\,C})$$

```
> eta1(t)=P[3,1];
```

$$\eta1(t) = -C2\,\mathrm{D}(\xi_1)(t) - C2\,R2\,\alpha\,\mathrm{D}(\xi_1)(t) - \alpha\,\xi_1(t)$$

```
> eta2(t)=P[4,1];
```

$$\eta2(t) = -\alpha\,\xi_1(t - \sqrt{L\,C}) + C2\,\mathrm{D}(\xi_1)(t - \sqrt{L\,C}) - C2\,R2\,\alpha\,\mathrm{D}(\xi_1)(t - \sqrt{L\,C})$$

```
> E(t)=P[5,1];
```

$$\mathrm{E}(t) = C2\,R2\,\alpha\,R1\,C1\,(\mathrm{D}^{(2)})(\xi_1)(t) - C2\,R1\,\alpha\,\%1 + C2\,R2\,\alpha\,\mathrm{D}(\xi_1)(t)$$
$$+ C2\,R1\,C1\,(\mathrm{D}^{(2)})(\xi_1)(t) + \alpha\,\xi_1(t) - 3\,\alpha^2\,R1\,\xi_1(t - 2\,\sqrt{L\,C}) + \alpha^2\,R1\,\xi_1(t)$$
$$+ C2\,\mathrm{D}(\xi_1)(t) + \alpha\,R1\,C1\,\mathrm{D}(\xi_1)(t) + C2\,R2\,\alpha^2\,R1\,\mathrm{D}(\xi_1)(t) + C2\,R1\,\alpha\,\mathrm{D}(\xi_1)(t)$$
$$- C2\,\%1 + \alpha\,\xi_1(t - 2\,\sqrt{L\,C}) - C2\,R1\,C1\,(\mathrm{D}^{(2)})(\xi_1)(t - 2\,\sqrt{L\,C}) + C2\,R2\,\alpha\,\%1$$
$$+ C2\,R2\,\alpha^2\,R1\,\%1 + C2\,R2\,\alpha\,R1\,C1\,(\mathrm{D}^{(2)})(\xi_1)(t - 2\,\sqrt{L\,C}) + \alpha\,R1\,C1\,\%1$$
$$\%1 := \mathrm{D}(\xi_1)(t - 2\,\sqrt{L\,C})$$

Therefore, we have parametrized the system variables *v1*, *v2*, $\eta1$, $\eta2$ and E by means of an arbitrary function ξ_1. Let us check whether or not ξ_1 can be expressed by means of the system variables *v1*, *v2*, $\eta1$, $\eta2$ and E. In order to do that, we need to determine whether or not the *Alg*-module associated with R is projective. This fact is equivalent to the vanishing of the second extension module ext^2 with values in *Alg* of the *Alg*-module associated with *R_ adj*.

```
> Ext2 := Exti(R_adj, Alg, 2);
```

$$Ext2 := [\begin{bmatrix} \delta \\ C2\,\mathrm{D} + C2\,\mathrm{D}\,R2\,\alpha + \alpha \end{bmatrix}, [\ 1\ ], \mathrm{SURJ}(1)]$$

As the first matrix *Ext2*[1] is not the identity matrix, we deduce that the *Alg*-module associated with R is not projective, and thus, not free. In particular, we conclude that the linear system is not flat. Let us compute the obstruction of flatness.

```
> PiPolynomial(R, Alg, [delta]);
```

$$[\delta]$$

Hence, the linear system is δ-flat, meaning that if we use the time-advance operator δ^{-1}, then the system becomes flat. In particular, using the time-advance operator δ^{-1}, we obtain the flat output $\xi = S\,(v1 : v2 : \eta 1 : \eta 2 : E)^T$ of the system, where S is the following matrix:

```
> S := LocalLeftInverse(Ext1[3], [delta], Alg);
```

$$S := \begin{bmatrix} 0 & -\frac{1}{2\,\delta\,\alpha} & 0 & 0 & 0 \end{bmatrix}$$

More precisely, we have:

```
> xi(t)=ApplyMatrix(S, [v1(t),v2(t),eta1(t),eta2(t),E(t)], Alg)[1,1];
```

$$\xi(t) = -\frac{1}{2}\,\frac{\mathrm{v2}(t + \sqrt{L\,C})}{\alpha}$$

Therefore, we obtain that *v2* is the δ-flat output of the linear system. Let us express the system variables *v1*, *v2*, $\eta 1$, $\eta 2$ and E in terms of the δ-flat output *v2*. In order to do that, we substitute the expression of $\xi(t)$ in terms of $\mathrm{v2}(t)$ into the parametrization of *v1*, *v2*, $\eta 1$, $\eta 2$ and E obtained above. Finally, we obtain:

```
> F := simplify(Mult(Ext1[3], S, Alg));
```

$$F :=$$

$$\left[0\,,\ \frac{C2\,\mathrm{D} + C2\,\mathrm{D}\,R2\,\alpha + \alpha - \mathrm{D}\,C2\,\delta^2 + R2\,\alpha\,\mathrm{D}\,C2\,\delta^2 + \delta^2\,\alpha}{2\,\alpha\,\delta}\,,\ 0\,,\ 0\,,\ 0\right]$$

$$[0\,,\ 1\,,\ 0\,,\ 0\,,\ 0]$$

$$\left[0\,,\ \frac{C2\,\mathrm{D} + C2\,\mathrm{D}\,R2\,\alpha + \alpha}{2\,\alpha\,\delta}\,,\ 0\,,\ 0\,,\ 0\right]$$

$$\left[0\,,\ \frac{-C2\,\mathrm{D} + C2\,\mathrm{D}\,R2\,\alpha + \alpha}{2\,\alpha}\,,\ 0\,,\ 0\,,\ 0\right]$$

$$\Big[0\,, -(\mathrm{D}^2\,C2\,R2\,\alpha\,R1\,C1 - C2\,\mathrm{D}\,\delta^2\,R1\,\alpha + C2\,\mathrm{D}\,R2\,\alpha + \mathrm{D}^2\,C2\,R1\,C1 + \alpha - 3\,\alpha^2\,\delta^2\,R1 + \alpha^2\,R1 + C2\,\mathrm{D} + \alpha\,R1\,C1\,\mathrm{D} + \mathrm{D}\,C2\,R2\,\alpha^2\,R1 + \mathrm{D}\,C2\,R1\,\alpha - \mathrm{D}\,C2\,\delta^2 + \delta^2\,\alpha - C2\,\mathrm{D}^2\,\delta^2\,R1\,C1 + R2\,\alpha\,\mathrm{D}\,C2\,\delta^2 + C2\,\mathrm{D}\,R2\,\alpha^2\,\delta^2\,R1 + C2\,\mathrm{D}^2\,R2\,\alpha\,\delta^2\,R1\,C1 + \alpha\,\delta^2\,R1\,C1\,\mathrm{D})/(2\,\alpha\,\delta),\ 0\,,\ 0\,,\ 0\Big]$$

In terms of the system variables, we have the following parametrization of *v1*, *v2*, $\eta 1$, $\eta 2$ and E by means of *v2*:

```
> v1(t)=ApplyMatrix(F, [v1(t),v2(t),eta1(t),eta2(t),E(t)], Alg)[1,1];
```

$$\mathrm{v1}(t) = \frac{1}{2}\,\frac{C2\,\mathrm{D}(v2)(t+\sqrt{L\,C})}{\alpha} + \frac{1}{2}\,C2\,R2\,\mathrm{D}(v2)(t+\sqrt{L\,C}) + \frac{1}{2}\,\mathrm{v2}(t+\sqrt{L\,C}) - \frac{1}{2}\,\frac{C2\,\mathrm{D}(v2)(t-\sqrt{L\,C})}{\alpha} + \frac{1}{2}\,C2\,R2\,\mathrm{D}(v2)(t-\sqrt{L\,C}) + \frac{1}{2}\,\mathrm{v2}(t-\sqrt{L\,C})$$

```
> v2(t)=ApplyMatrix(F, [v1(t),v2(t),eta1(t),eta2(t),E(t)], Alg)[2,1];
```

$$\mathrm{v2}(t) = \mathrm{v2}(t)$$

```
> eta1(t)=ApplyMatrix(F, [v1(t),v2(t),eta1(t),eta2(t),E(t)], Alg)[3,1];
```

$$\eta 1(t) = \frac{1}{2}\,\mathrm{v2}(t+\sqrt{L\,C}) + \frac{1}{2}\,\frac{(C2 + C2\,R2\,\alpha)\,\mathrm{D}(v2)(t+\sqrt{L\,C})}{\alpha}$$

```
> eta2(t)=ApplyMatrix(F, [v1(t),v2(t),eta1(t),eta2(t),E(t)], Alg)[4,1];
```

$$\eta 2(t) = \frac{1}{2}\,\mathrm{v2}(t) + \frac{1}{2}\,\frac{(-C2 + C2\,R2\,\alpha)\,\mathrm{D}(v2)(t)}{\alpha}$$

```
> E(t)=ApplyMatrix(F, [v1(t),v2(t),eta1(t),eta2(t),E(t)], Alg)[5,1];
```

$$\mathrm{E}(t) = -\frac{1}{2}\,C2\,R2\,R1\,C1\,(\mathrm{D}^{(2)})(v2)(t+\sqrt{L\,C}) + \frac{1}{2}\,C2\,R1\,\%1 - \frac{1}{2}\,C2\,R2\,\%2 - \frac{1}{2}\,\frac{C2\,R1\,C1\,(\mathrm{D}^{(2)})(v2)(t+\sqrt{L\,C})}{\alpha} - \frac{1}{2}\,\mathrm{v2}(t+\sqrt{L\,C}) + \frac{3}{2}\,\alpha\,R1\,\mathrm{v2}(t-\sqrt{L\,C}) - \frac{1}{2}\,\alpha\,R1\,\mathrm{v2}(t+\sqrt{L\,C}) - \frac{1}{2}\,\frac{C2\,\%2}{\alpha} - \frac{1}{2}\,R1\,C1\,\%2 - \frac{1}{2}\,C2\,R2\,\alpha\,R1\,\%2 - \frac{1}{2}\,C2\,R1\,\%2 + \frac{1}{2}\,\frac{C2\,\%1}{\alpha} - \frac{1}{2}\,\mathrm{v2}(t-\sqrt{L\,C}) + \frac{1}{2}\,\frac{C2\,R1\,C1\,(\mathrm{D}^{(2)})(v2)(t-\sqrt{L\,C})}{\alpha} - \frac{1}{2}\,C2\,R2\,\%1 - \frac{1}{2}\,C2\,R2\,\alpha\,R1\,\%1 - \frac{1}{2}\,C2\,R2\,R1\,C1\,(\mathrm{D}^{(2)})(v2)(t-\sqrt{L\,C}) - \frac{1}{2}\,R1\,C1\,\%1$$

$$\%1 := \mathrm{D}(v2)(t-\sqrt{L\,C})$$

$$\%2 := \mathrm{D}(v2)(t+\sqrt{L\,C})$$

Now, let us return to the non-linear system defined above. The previous computations of the parametrization show that the non-linear system is triangular:

1. From the second equation of the non-linear system, we can express $\eta 1$ in terms of *v2* by using a derivation and an advance operator δ^{-1}.

2. From the fourth equation, we can express $\eta 2$ in terms of *v2* and $\eta 1$, and thus, in terms of *v2* only by substituting the previous expression of $\eta 1$ in terms of *v2*.

3. From the third equation, we can express *v1* in terms of $\eta 1$ and $\eta 2$, and thus, in terms of *v2* only by using the expressions already obtained.

4. From the first equation, we obtain E in terms of *v1* and *v2*, and thus, in terms of *v2* only.

Therefore, the non-linearity $R1\,\mathrm{f1}(\mathrm{v1}(t))$ will only appear in step 4 when we express E in terms of *v1*. Hence, the parametrization of $\eta 1$, $\eta 2$ and *v1* in terms of *v2* is the same as in the linear model. More precisely, by sustituting $\alpha = \sqrt{\frac{C}{L}}$, we have:

```
> v1(t)=subs(alpha=sqrt(C/L), ApplyMatrix(F,[v1(t),v2(t),eta1(t),eta2(t),E(t)],
> Alg)[1,1]);
```

$$\mathrm{v1}(t) = \frac{1}{2}\,\frac{C2\,\mathrm{D}(v2)(t+\sqrt{L\,C})}{\sqrt{\frac{C}{L}}} + \frac{1}{2}\,C2\,R2\,\mathrm{D}(v2)(t+\sqrt{L\,C}) + \frac{1}{2}\,\mathrm{v2}(t+\sqrt{L\,C})$$
$$-\frac{1}{2}\,\frac{C2\,\mathrm{D}(v2)(t-\sqrt{L\,C})}{\sqrt{\frac{C}{L}}} + \frac{1}{2}\,C2\,R2\,\mathrm{D}(v2)(t-\sqrt{L\,C}) + \frac{1}{2}\,\mathrm{v2}(t-\sqrt{L\,C})$$

```
> v2(t)=ApplyMatrix(F, [v1(t),v2(t),eta1(t),eta2(t),E(t)],Alg)[2,1];
```

$$\mathrm{v2}(t) = \mathrm{v2}(t)$$

```
> eta1(t)=subs(alpha=sqrt(C/L),ApplyMatrix(F,[v1(t),v2(t),eta1(t),eta2(t),E(t)],
> Alg)[3,1]);
```

$$\eta 1(t) = \frac{1}{2}\,\mathrm{v2}(t+\sqrt{L\,C}) + \frac{1}{2}\,\frac{(C2 + C2\,R2\,\sqrt{\frac{C}{L}})\,\mathrm{D}(v2)(t+\sqrt{L\,C})}{\sqrt{\frac{C}{L}}}$$

```
> eta2(t)=subs(alpha=sqrt(C/L), ApplyMatrix(F,[v1(t),v2(t),eta1(t),eta2(t),E(t)],
> Alg)[4,1]);
```

$$\eta 2(t) = \frac{1}{2}\,\mathrm{v2}(t) + \frac{1}{2}\,\frac{(-C2 + C2\,R2\,\sqrt{\frac{C}{L}})\,\mathrm{D}(v2)(t)}{\sqrt{\frac{C}{L}}}$$

Finally, after having expressed E in terms of *v1*, we can substitute the expression of *v1* in terms of *v2* in order to obtain the following parametrization of E:

```
> E(t)=subs(alpha=sqrt(C/L), ApplyMatrix(F[5,2], [v2(t)], Alg)[1]
> +R1*f1(ApplyMatrix(F[1,2], [v2(t)], Alg)[1]));
```

$$\mathrm{E}(t) = -\frac{1}{2}\,C2\,R2\,R1\,C1\,(\mathrm{D}^{(2)})(v2)(t+\sqrt{L\,C}) + \frac{1}{2}\,C2\,R1\,\%1 - \frac{1}{2}\,C2\,R2\,\%2$$
$$-\frac{1}{2}\,\frac{C2\,R1\,C1\,(\mathrm{D}^{(2)})(v2)(t+\sqrt{L\,C})}{\sqrt{\frac{C}{L}}} - \frac{1}{2}\,\mathrm{v2}(t+\sqrt{L\,C}) + \frac{3}{2}\,\sqrt{\frac{C}{L}}\,R1\,\mathrm{v2}(t-\sqrt{L\,C})$$
$$-\frac{1}{2}\,\sqrt{\frac{C}{L}}\,R1\,\mathrm{v2}(t+\sqrt{L\,C}) - \frac{1}{2}\,\frac{C2\,\%2}{\sqrt{\frac{C}{L}}} - \frac{1}{2}\,R1\,C1\,\%2 - \frac{1}{2}\,C2\,R2\,\sqrt{\frac{C}{L}}\,R1\,\%2$$
$$-\frac{1}{2}\,C2\,R1\,\%2 + \frac{1}{2}\,\frac{C2\,\%1}{\sqrt{\frac{C}{L}}} - \frac{1}{2}\,\mathrm{v2}(t-\sqrt{L\,C}) + \frac{1}{2}\,\frac{C2\,R1\,C1\,(\mathrm{D}^{(2)})(v2)(t-\sqrt{L\,C})}{\sqrt{\frac{C}{L}}}$$
$$-\frac{1}{2}\,C2\,R2\,\%1 - \frac{1}{2}\,C2\,R2\,\sqrt{\frac{C}{L}}\,R1\,\%1 - \frac{1}{2}\,C2\,R2\,R1\,C1\,(\mathrm{D}^{(2)})(v2)(t-\sqrt{L\,C})$$
$$-\frac{1}{2}\,R1\,C1\,\%1 + R1\,\mathrm{f1}\Bigg(\frac{1}{2}\,\frac{C2\,\%2}{\sqrt{\frac{C}{L}}} + \frac{1}{2}\,C2\,R2\,\%2 + \frac{1}{2}\,\mathrm{v2}(t+\sqrt{L\,C}) - \frac{1}{2}\,\frac{C2\,\%1}{\sqrt{\frac{C}{L}}}$$
$$+\frac{1}{2}\,C2\,R2\,\%1 + \frac{1}{2}\,\mathrm{v2}(t-\sqrt{L\,C})\Bigg)$$
$$\%1 := \mathrm{D}(v2)(t-\sqrt{L\,C})$$
$$\%2 := \mathrm{D}(v2)(t+\sqrt{L\,C})$$